

Graphics

A complex drawing in descriptive geometry

Denis Roegel

Abstract

This article describes the reproduction of a complex drawing in descriptive geometry. The original plate was published by Théodore Olivier in 1842 and represents the meshing of two gears on skew axes. The drawing was analyzed and redone in METAPOST, and the construction illustrates a number of typical features of that programming environment.

1 Introduction

The French geometer Gaspard Monge (1746–1818) developed “descriptive geometry”, and this discipline flourished during the 19th century, at the same time as the industrial revolution, and the construction of machines, buildings, and other masterpieces of architecture.

Geometry, and in particular descriptive geometry, is a wonderful application for METAPOST enthusiasts (Goossens, Mittelbach, Rahtz, Roegel, and Voß, 2007). METAPOST makes it possible to draw lines and curves in a very exact way, and at the same time relate different parts of a drawing to each other, as they should be in descriptive geometry.

In this article, I go into the details of the construction of a complex drawing, taken from the work of Théodore Olivier (1793–1853), one of Monge’s best followers. Olivier was a former student of the *École Polytechnique* and went on to do groundbreaking work in the geometrical theory of gears. In 1829 he was one of the founders of the *École Centrale des Arts et Manufactures*. One of his most important books is his *Théorie géométrique des engrenages* (Olivier, 1842), where exact drawings for gearings are produced in an almost purely geometrical way. Application of this theory led to the many models found in the *Musée des arts et métiers* in Paris.

One of the chapters in his book is devoted to the meshing of wheels with non-intersecting axes, and my purpose is to show how Olivier’s corresponding drawing (figure 1) can be produced with a tool such as METAPOST. But this article is neither meant as an introduction to the theory of gears, nor to the rules of descriptive geometry. I will focus only on geometrical relations, without always stating why things are so or not. I am taking the vantage point of an engineer who has some drawing to produce, of

which he/she knows the geometrical relationships, but the grounds for these relationships will not be essential in our analysis. The reader interested in more details may consult Olivier’s book or other books on the theory of gears.

Figure 2 shows the final figure produced with our code.

2 Olivier’s plate

Olivier writes (Olivier, 1842, p. 118) that the plate represents the original-scale working-drawing which was used to manufacture the gearing-model to transmit rotation motion between two axes not in the same plane and having an angle of 30° between them.

LT (figure 2) represents the “ground line” (*ligne de terre*). The axis A of the wheel — carrying twenty-four cylindrical teeth with circle involute shapes — is vertical; its projections are A^h and A^v .

The axis A_1 of the wheel — carrying eighteen helical teeth — is in the vertical projection plane; its projections are A_1 and A_1^h or LT .

The middle lines M and M_1 are, on the vertical projection plane, the vertical traces of the two planes dividing the rings to cut in equal parts.

These lines M and M_1 intersect at X^v which is the vertical projection of the line X , intersection of the two planes dividing the rings in equal parts.

The wheel attached to the inclined axis A_1 was turned around the line X , in order to bring this wheel into a horizontal position.

It is in this horizontal position that the wheel carrying eighteen teeth is represented by the working drawing.

T is a vertical plane tangent to the cylinders H and H_1 . The vertical line Y^v is in the plane T .

For each wheel, the outer circle shows the base circle of the involutes. The inner circle shows how far the teeth from the other wheel penetrate one wheel.

Olivier’s plate is also reproduced in von Seherr-Thoss’s book on the development of gearing technology (von Seherr-Thoss, 1965, p. 120), but the drawing was redone and some errors were not corrected (figure 3).

3 Involute cylindrical and helical teeth

3.1 Involute teeth

Figure 4 shows the construction of the involute curve of a circle. It is straightforward to obtain its cartesian equations which are:

$$\begin{aligned}x_M &= r(\cos \theta + \theta \sin \theta) \\y_M &= r(\sin \theta - \theta \cos \theta)\end{aligned}$$

with θ being expressed in radians.




Figure 1: Olivier's plate, with a frame measuring 35cm \times 42.5cm (Olivier, 1842). Note that a tooth is missing on the lower wheel.

The following METAPOST code returns a path for an involute tooth. The three parameters of the macro `involute_tooth` are the radius r of the base circle, an angle a for truncating the involute, and the angular step s for θ . It turns out that $s = 20$ gives an excellent approximation of the involute curve.

```
RAD=3.14159/180; % conversion degrees -> radians
vardef involute_tooth(expr r,a,s)=
  save p,t;
  path p;
  p=(r,0)
  for i=1 step s until 90:
    ..(r*(cosd(i) + RAD*i*sind(i)),
      r*(sind(i) - RAD*i*cosd(i)))
  endfor;
  t=xpart(p intersectiontimes
    (origin--(3r*dir(a))));
  (subpath(0,t) of p)
enddef;
```

Two more macros are defined for the linear parts of the cylindrical teeth of the lower wheel:

```
vardef cyl_full_tooth(expr r,a,b)=
  save p,L;path p;pair L;
  p=involute_tooth(r,a,20);
  L=point length(p) of p;
  (p--(L rotated b)--(r*dir(a+b)))
enddef;
vardef cyl_full_tooth_x(expr r,a,b,c)=
  save p,L;
  path p;pair L;
```

```
p=involute_tooth(r,a,20);
L=point length(p) of p;
((L rotated b)--(r*dir(a+b+c)))
enddef;
```

3.2 Tooth contact

Teeth can be put in contact easily, provided the shapes of the curves to put in contact and the point of contact are known. We then only needed to find out how much the standard teeth curves need to be rotated.

In the example below, two involutes p_1 and p_2 are defined, and the angles a and b (from the centers of the respective circles) between the horizontal and the contact of the involutes with circles corresponding to the final contact are determined. Then the two involutes are merely rotated according to these two angles (see figure 5):

```
p1=involute_tooth(v,25,5);p2=p1 shifted z2;
a=angle(p1 intersectionpoint
  circle(z1,.5arlength(z1--z2)));
b=angle((p2 intersectionpoint
  circle(z2,.5arlength(z1--z2)))-z2);
draw p1 rotated -a;
draw p2 rotatedaround(z2,180-b);
```




Figure 2: The entire figure (except the ancillary drawings within the frame) reproduced from Olivier’s plate.

3.3 Helical teeth with involute profiles

The teeth in Olivier’s upper (helical) wheel are obtained from the envelope of the tangents to a helix (see figure 6). The intersections of the envelope with planes orthogonal to the cylinder axis are circle involutes. This can be proved easily: let the helix equation be

$$M = \begin{pmatrix} r \cos \theta \\ r \sin \theta \\ a\theta \end{pmatrix}$$

A tangent vector to the helix in M is

$$\frac{dM}{d\theta} = \begin{pmatrix} -r \sin \theta \\ r \cos \theta \\ a \end{pmatrix}$$

A point $T(\alpha)$ of the tangent has the coordinates

$$T(\alpha) = \begin{pmatrix} r \cos \theta - \alpha r \sin \theta \\ r \sin \theta + \alpha r \cos \theta \\ a\theta + \alpha a \end{pmatrix}$$

Taking $a\theta + \alpha a = 0$, we obtain $\alpha = -\theta$ and therefore the intersection with the plane $z = 0$ is

$$I(\theta) = \begin{pmatrix} r(\cos \theta + \theta \sin \theta) \\ r(\sin \theta - \theta \cos \theta) \\ 0 \end{pmatrix}$$




Figure 3: Seherr-Thoss's reproduction of Olivier's drawing (von Seherr-Thoss, 1965, p. 120). The most obvious changes are the German labels, but actually, as a close examination shows, the whole drawing was redone.




Figure 4: The construction of an involute.


which is exactly the equation of the involute described above. The intersection with any other plane $z = h$ also gives an involute, of course.

4 The construction


Olivier's drawing is reconstructed as follows:

4.1 Units

All the dimensions are expressed as multiples of a conventional unit, so that it becomes easy to scale the figure afterwards. In this case, the unit u was 1 cm on the original drawing. Changing the unit still



(a) Two involute teeth.



(b) The two rotated teeth meeting in O .

Figure 5: The contact of two involute teeth.




Figure 6: Helical tooth with involute profiles. The tooth is generated by an involute \mathcal{I} moving along a helix ξ . (In this drawing, the four non-working faces of the tooth have been made planar.)

affects the drawing, in that text and line widths are usually not changed when the unit is changed (but they could be taken into account).

```
numeric u;
u=5mm;
```

4.2 Useful macros

A few useful macros are defined, in particular a shortcut for a circle of center c and radius r :




Figure 7: Step 1. The distance $D = z_0 - z_{200}$ is the shortest distance between the axes A and A_1 .

```
def circle(expr c,r)=
  (fullcircle scaled 2r shifted c)
endef;

% pathpoint(p) is some point on
% (point 0 of p, point 1 of p)
def pathpoint(text p)=
  whatever[point 0 of p,point 1 of p]
endef;
```

4.3 General layout

We start with the general layout of the figure, and its various parameters. The number of teeth of the lower (cylindrical) wheel is $n_a = 24$, whereas the number of teeth of the upper (helical) wheel is $n_b =$

18. The angle between the two wheel axes is $\beta = 30^\circ$. The (shortest) distance between the two axes is d_0 .

On figure 7, the “ground line” is p_0 ; this separates the horizontal projection (bottom) from the vertical projection (top). Below the p_0 line, we see two wheels, one with a vertical axis A (located at z_0), and another with an inclined axis A_1 . However, on the horizontal projection, the wheel was first turned, so that its axis in fact is also vertical. The base cylinders of the two wheels therefore appear as circles on the horizontal projection.

p_0 is also the (horizontal) projection of the A_1

axis and point z_{200} is the projection of the point on A_1 which is at the closest distance from A .

```
na=24;    % number of teeth of wheel 0
nb=18;    %                               1
beta=30;  % angle between the wheel axes
z0=origin; % center of wheel 0
d0=23.3u; % distance between the axes of the wheels
z200-z0=d0*up;
```

The two cylinders H (lower wheel) and H_1 (upper wheel) are provided with teeth and these teeth are positioned on two bands. The (vertical) projections of the teeth for H are located between p_0 and p_7 , whereas those for H_1 are located between p_4 and p_5 . The position of these bands determines the contact between the two wheels and is a parameter of the drawing. The intersection of the median planes of these bands is a line X whose vertical projection is $z_{201} = X^v$. We use d_1 and d_2 as parameters locating X^v . The line p_0 can now be defined.

```
d1=2.9u; % distance from X to the line LT
d2=4.55u; % distance from X to the vertical A^v
z201-z200=(d2,d1);z210-z200=d1*up;
z203=z200+6u*left;
z204=z200+25u*right;p0=z203--z204; % LT line
```

The radii r_1 and r_2 of the two cylinders depend both on the number of teeth and on the angle between the two axes. These values will serve as base radii for the teeth's involute curves, but teeth will be undercut below these base radii for deeper meshing. The undercut radii are r_{11} and r_{21} . Finally, z_1 is the intersection between X and the tangency plane common to the two cylinders H and H_1 . The (horizontal) projection of this tangency plane T is p_1 .

```
r1=d0/(1+(nb/na)/cosd(beta)); % radius of wheel 0
r2=d0-r1; % radius of wheel 1
r11=11.7u; % inner radius of wheel 0
r21=9.6u; % inner radius of wheel 1
```

```
z1=(x201,r1);
z2=z0+r1*up+15.2u*left;
z3=z0+r1*up+22u*right;
p1=z2--z3; % tangent between the wheels
```

The upper wheel is rotated around X by an angle β and the horizontal projection of its center is z_{100} . The intersection of p_3 and A_1 is z_{220} . The (vertical) projection of the point on A_1 closest to A is z_{219} . Finally, we compute z_{202} and z_{207} , a segment parallel to A_1 and going through X^v .

```
z100=whatever[z0,z1]
=whatever[z200,z200+right];
z220=z201+(x100-x201)*dir(beta);
z219=whatever[z200,z210]
=whatever[z220,z220+dir(90+beta)];
z202=whatever[z201,z201+z219-z220]
=whatever[z210,z219];
z207=whatever[z202,z201]
=whatever[z200,z100];
```

Two contours p_{101} and p_{102} are now determined for the purpose of framing or clipping parts of the drawing under construction: p_{101} frames the whole drawing, and p_{102} frames the upper part. These contours will be used to hide parts of the teeth, as in Olivier's original plate.

```
% contours:
z301=z0+1.3r1*left;
z302=(xpart(point 1 of p0),y0);
z303=(x302,y100+1.3r2);
z304=(x301,y303);
% frame for the whole drawing:
p101=z301--z302--z303--z304--cycle;
z251=(x0,y303); % vertical edge
% contour for clipping the upper teeth:
z305=(x303,y100);
z306=(x304,y100);
p102=z301--z302--z305--z306--cycle;
```

The lines M , M_1 (dividing the teeth bands) and their parallels are now easily defined:

```
p2=(xpart(point 0 of p0),y210)
--(z201+4(z201-z210)); % M
p3=(z220+2(z220-z201))--(z201-2(z220-z201)); % M1
p4=p3 shifted (d1*dir(90+beta)); % parallel to M1
p5=p3 shifted (d1*dir(-90+beta)); % parallel to M1
p7=p2 shifted (d1*up);
```

Next we define p_6 (the A_1 axis) using z_{221} and z_{222} . This axis being partly drawn dashed, we introduce two intermediate points z_{223} and z_{223} .

```
z221=.4[z219,z220];z222=(z220+.65(z220-z219));
p6=z221--z222; % A1 axis
z223=p6 intersectionpoint p4;
z224=p6 intersectionpoint p5;
```

The projection of the rotated axis A_1 is p_8 :

```
z101-z100=z100-z102=r2*up;
p8=z101--z102;
```

The contacts between the helical teeth of the upper wheel and the cylindrical teeth of the lower wheel are vertical segments which are parallel to A , and one of them is the segment (z_{228}, z_{233}) . Since the upper wheel is rotated for the drawing, the segments need to be rotated too. A number of arcs on figure 7 show the points and their position after rotation. For instance, z_{223} (intersection of axis A_1 and p_4) becomes z_{225} . For the main contact between the wheels, we define points z_{226} (p), z_{227} (p'), z_{228} (i^v), z_{229} (q'), z_{230} (q), z_{231} (i'^v), z_{232} (l'^v) and z_{233} (l^v). The segment $p_9 = l'^v - i'^v$ is the vertical projection of the main rotated contact.

The vertical line through X^v is tangent to one of the cylindrical teeth. The next vertical tangent to the right is at a distance which is the circumference of the lower wheel divided by n_a . The intersection with LT is z_{234} . The contact segment on the upper band is (z_{236}, z_{238}) . The horizontal projection of that contact is point z_{12} on the cylinder tangency line.




Figure 8: Excerpt of Olivier's plate: contact segments. The arc at the lower right corner seems incorrect.

```
z225=p8 intersectionpoint p7;

% rotation arcs
z226=z201+d1*dir(90+beta); % p
z227=z201+d1*up; % p'
z228=whatever[z201,z201+up]
=whatever[z226,z226+dir(beta)]; % i^v
z229=z201+d1*down; % q'
z230=z201+d1*dir(-90+beta); % q
z231=z228 rotatedaround(z201,-beta); % i^v
z232-z201=z201-z231; % 232=l^v
z233-z201=z201-z228; % 233=l^v
p9=z232--z231; % l^v - i^v
```

At a given time, several teeth are in contact. In our case, three teeth are in contact. We call these teeth 1, 2 and 3. The segment p_9 is the contact for the second lower tooth.

```
z234=z229+(2*3.14159*r1/na)*right;
% intersection with p4 (=upper line parallel to M1)
z236=pathpoint(p4)=z234+whatever*up;
% intersection with p5 (=lower line parallel to M1)
z238=(z236--z234) intersectionpoint p5;
z12=(x236,y0+r1); % contact with lower tooth
```

p_{10} is the rotated contact segment for the third tooth. It is obtained by translating the second contact segment.

```
z235=z236 rotatedaround(z201,-beta); % end of arc
% parallel to l^v-i^v:
p10=p9 shifted (z235-z231);
z239=point 0 of p10; % p10=z239--z235
```

z_{238} is used to define the two lines p_{11} and p_{12} .

```
z241=(z236--z12) intersectionpoint p2;
```




Figure 9: Step 2. Involutes tangent to the vertical lines going through z_{111}, \dots, z_{116} .

```
% parallel to M going through 'a':
p11=p2 shifted (z241-z238);
% parallel to M going through 'b':
p12=p2 shifted (z238-z241);
```

z_{211} and z_{212} are the intersections of p_{11} and p_{12} with the vertical projection of A :

```
z211=p11 intersectionpoint (z0--z219); % a
z212=p12 intersectionpoint (z0--z219); % b
```

The vertical tangent to the first tooth, going through z_{242} , is obtained from the two other tangents (going through z_{229} and z_{234}). z_{11} is the contact with the first tooth on the tangency line T .

```
z242=z229-(z234-z229);
% intersection with p11:
z243=(x242,y241+(y241-y238));
z11=(x242,y0+r1);
p13=z11--z243; % vertical tangent to lower tooth 1
```

Next we rotate the contact segment and obtain p_{14} .

```
z244=z243 rotatedaround(z201,-beta);
% parallel to l^v-i^v:
p14=p9 shifted (z244-z231);
z245=point 0 of p14; % p14=z245--z244
z246=z245 rotatedaround(z201,beta);
```

4.4 Tangents to the lower teeth

The three contact segments have the horizontal projections z_{11} , z_1 and z_{12} . The intersections of these segments with the planes whose (vertical) projections are p_4 and p_5 are located on involutes, as shown in figure 6. These involutes are actually tangent to a line orthogonal to the (horizontal) projection of the contact segment, as one can easily see. The six involutes under consideration go through the points z_{111}, \dots, z_{116} (figure 9).

```
% tangent 1 with upper tooth 1:
z111=(x245,y0+r1);p15=z245--z111;
z112=(x244,y0+r1);p16=z244--z112; % tang. 2/tooth 1
z113=(x232,y0+r1);p17=z232--z113; % tang. 1/tooth 2
z114=(x231,y0+r1);p18=z231--z114; % tang. 2/tooth 2
z115=(x239,y0+r1);p19=z239--z115; % tang. 1/tooth 3
z116=(x235,y0+r1);p20=z235--z116; % tang. 2/tooth 3
```



```
z106=p53 intersectionpoint circle(z100,d5);
p54=p53 rotatedaround
(z100,angle(z112-z100)-angle(z106-z100));
```

The two involutes p_{52} and p_{54} extend below the tangency line, and we cut them so that they end at that line. We now have three edges of the projected tooth: p_{55} , p_{56} and p_{57} .

```
p55=p52 cutafter circle(z100,d4);
p56=p54 cutafter circle(z100,d5);
p57=z111--z112;
```

The other faces of the tooth are non-working, and they can be determined in different ways, taking into account the way they are manufactured and the material used. The tooth thickness is a parameter of the drawing and we determine z_{109} by rotating z_{111} of a certain angle.

```
z109-z100=(z111-z100) rotated 2.5; % empirical
```

z_{122} is obtained as an intersection of a circle going through z_{112} and a line parallel to p_{57} and going through z_{109} . However, although this makes it easy to draw the figure, it produces a face which is not planar. Since Olivier did so in his plate, we reproduced it. In figure 6, however, we made sure the corresponding face is planar.

```
p58=z111--z109;
z110-z112=z109-z111;
z121-z110=z110-z109;
z122=(z109--z121) intersectionpoint circle(z100,d5);
p59=z109--z122;
p60=z112--z122;
```

z_{124} is obtained as the intersection of the base circle with a parallel to (z_{109}, z_{100}) going through z_{122} . As a consequence, the face which is opposite the working-face can be made planar.

```
z123=point 0 of p56;
z124=(z122--(z122+(z100-z109)))
intersectionpoint circle(z100,r2);
p61=z122--z124;
```

And when the wheels turn, the contact segment shifts from the outside to the inside of the tooth, or in the opposite direction. When the contact occurs at the base radius, the contact segment which is involved is the segment (z_{123}, z_{126}) , and z_{126} is computed as follows:

```
z126=p55 intersectionpoint
(z123--(z123+(z100-z123) rotated 90));
p62=z123--z126;
```

We now cut the tips of the helical teeth, in order to limit their projection inside the cylindrical wheel. We compute a radius d_6 based on the value of r_{11} measured on Olivier's plate. p_{63} is the circle for the maximum extent of the helical teeth.

```
r10=arclength(z0--z100);
d6=.99(r10-r11);
p63=circle(z100,d6);
```

Using the previous circle, we cut a part of the helical tooth. For that, we determine three points:

```
z127=p55 intersectionpoint p63;
z128=p57 intersectionpoint p63;
z129=p59 intersectionpoint p63;
```

Then, we define several paths. p_{64} is the arc going from z_{127} to z_{129} . p_{65} is the part of p_{55} that is left once we cut what goes beyond p_{63} . p_{66} and p_{67} are two more edges produced by this cut.

```
p64=the_arc(z127,z100,
angle(z129-z100)-angle(z127-z100));
p65=p55 cutafter p63;
p66=z128--z112;
p67=z129--z122;
```

The macro `the_arc` is defined as follows:

```
vardef the_arc(expr s,c,a)=
save p,t;
path p;
p=if a<0:reverse fi
fullcircle rotated (angle(s-c))
scaled (2arclength(s--c)) shifted c;
t=xpart(p intersectiontimes
(c--(c+2(s-c) rotated a)));
(subpath(0,t) of p)
enddef;
```

Several points are defined for the purpose of drawing the line from z_{100} to the tip z_{109} :

```
z130=(z109--z100) intersectionpoint
circle(z100,r21);
% dashed line to the tip of the upper tooth (1):
p68=z109--z130;
z131=(z109--z100) intersectionpoint p63;
z132=(z109--z100) intersectionpoint circle(z100,r2);
% dashed line to the tip of the upper tooth (2):
p69=z131--z132;
```

Finally, we construct a contour for the whole upper wheel (figure 12). This contour is made of the three paths p_{71} , p_{64} , p_{70} , and these paths are repeated to form the contour p_{73} :

```
% construction of a contour for drawing
% the dashed lines of the lower teeth:
p70=p67 cutafter (p55 rotatedaround(z100,(360/nb)));
p71=p65 cutbefore
(p67 rotatedaround(z100,-(360/nb)));
p72=p64--p70--(p71 rotatedaround(z100,(360/nb)));

% contour to hide the lower teeth:
p73=p72 for i=1 upto nb-1:
--(p72 rotatedaround(z100,i*(360/nb)))
endfor--cycle;
```

Four additional paths are defined for the variant teeth shown for the upper wheel in Olivier's plate:

```
p74=p65 cutafter
(p67 rotatedaround(z100,-(360/nb)));
p75=p55 cutbefore
(p67 rotatedaround(z100,-(360/nb)));
p76=p62 cutbefore
(p61 rotatedaround(z100,-(360/nb)));
p77=p62 cutafter
(p61 rotatedaround(z100,-(360/nb)));
```




Figure 12: Step 4. Construction of the upper wheel contour.

4.6 Construction of the lower wheel teeth

The lower teeth are easier to draw, since they are purely cylindrical (figure 13). They are positioned in the same way as the involutes for the upper teeth. In other words, the involutes are rotated in such a way that the contact occurs on z_1 . The teeth shapes are made of two paths, p_{22} and p_{24} , and the same technique is applied for both. Only the macro used is not the same.

```
p21=cyl_full_tooth(r1,5,2);
d3=arclength(z0--z1);
z14=p21 intersectionpoint circle(origin,d3);
p22=p21 rotated (angle(z1)-angle(z14));
% second part of the tooth:
p23=cyl_full_tooth_x(r1,5,2,1);
p24=p23 rotated (angle(z1)-angle(z14));
```

Finally, the lower teeth are partly hidden by drawing them all and then using the `hiddenpath` macro.

```
def draw_lower_teeth=
  for i=-5 upto 7: % Olivier forgot the case i=7
    draw p22 rotatedaround(origin,i*(360/na));
    draw p24 rotatedaround(origin,i*(360/na));
    % we remove what lies inside p73:
    hiddenpath(p22 rotatedaround(origin,i*(360/na)),
              p73,dashtype(1));
    hiddenpath(p24 rotatedaround(origin,i*(360/na)),
              p73,dashtype(1));
  endfor;
enddef;
```

The `hiddenpath` macro is defined as follows. `hiddenpath(under,over)(dt)` draws that part of

the path “*under*” which is within path “*over*” with dashes of type “*dt*”.

```
vardef hiddenpath(expr under,over)(text dt)=
  save p,q;
  picture p,q;
  p=image(draw under);clip p to over;
  undraw p;
  q=image(draw under dt);
  clip q to over;
  draw q;
enddef;
```

and the various dashes are obtained with:

```
def dashtype(expr n)=
  if n=0: dashed withdots
  elseif n=1: dashed evenly
  elseif n=2: dashed
    dashpattern(
      on 6bp off 2bp on 1bp off 2bp on 1bp off 2bp)
  elseif n=3: dashed
    dashpattern(on 6bp off 2bp on 1bp off 2bp)
  fi
enddef;
```

5 Drawing the figure

Given all the previous definitions, drawing the figure is pretty straightforward, and we won’t describe it in detail. We will merely give an insight into the hatched line *LT* and the clipping of parts of the drawing.

The hatched line *LT* is drawn with the macro:

```
vardef hatch(expr p,n,l)=
  save A,B;
  pair A,B;
  A=point 0 of p;B=point 1 of p;
```




Figure 13: Step 5. Lower wheel teeth construction, and dashing.

```
for i=0 upto n:
  draw (i/n)[A,B]--((i/n)[A,B]+1*dir(-45));
endfor;
enddef;
```

and 400 marks are produced with a call to

```
hatch(p0,400,.2u);
```

Clipping the upper wheel may be seen as tricky, especially if parts of the lower wheel have been previously drawn, and we want to retain them. The solution is to save the current picture in a `picture` variable, then to reset the current picture, to draw the upper wheel, then to clip it, and finally to redraw the saved picture. This is summarized in the following macro:

```
vardef draw_upper_wheel=
  save oldpic;
  picture oldpic;
  oldpic=currentpicture;
  currentpicture:=nullpicture;
  draw_upper_teeth;
  draw_upper_wheel_structure;
  clip currentpicture to p102; % we cut beyond p102
  draw oldpic;
enddef;
```

6 Conclusion

We were able to produce a very faithful copy of Olivier's original plate. We think that we identified two errors in Olivier's drawing (or rather in the etching which was made from his drawing). First, the wheel with cylindrical teeth is obviously missing

one tooth in Olivier's drawing. Second, the arc between points z_{224} and z_{100} is incorrectly represented.

Other than that, reproducing Olivier's figure has provided an understanding that would be difficult to reach by merely gazing at the drawing. By reproducing it, one is led to find the relationships between the parts, and, in particular, to try to minimize the number of parameters, such that the figure becomes as general as possible.

References

- Goossens, Michel, F. Mittelbach, S. Rahtz, D. Roegel, and H. Voß. *The L^AT_EX Graphics Companion, Second Edition*. Addison-Wesley, 2007.
- Olivier, Théodore. *Théorie géométrique des engrenages destinés à transmettre le mouvement de rotation entre deux axes situés ou non situés dans un même plan*. Paris: Bachelier, 1842.
- von Seherr-Thoss, Hans Christoph. *Die Entwicklung der Zahnrad-Technik: Zahnformen und Tragfähigkeitsberechnung*. Berlin: Springer-Verlag, 1965.

◇ Denis Roegel
 LORIA, BP 239
 54506 Vandœuvre-lès-Nancy
 FRANCE
 roegel (at) loria dot fr
<http://www.loria.fr/~roegel>